The date of Easter

Easter Day, which always occurs on a Sunday, is the day to which such moveable feasts as Whitsun and Trinity Sunday in the Christian calendar are fixed, and is defined in The Explanatory Supplement to the
Astronomical Almanac (1992) as follows:

> *In the Gregorian calendar, the date of Easter is defined to occur on the Sunday following the ecclesiastical full moon that falls on or next after March 21st.*

The problem is that the ecclesiastical full Moon is not the same as the astronomical full Moon. The former is based on a set of tables, which do not take into account the complexity of the Moon's motion. As a fair guide, we may say that Easter Day is usually the first Sunday after the fourteenth day after the first new Moon after 21 March. Several authors have provided algorithms for calculating the date of Easter. You can, for example, use the methods and tables given in the Book of Common Prayer (1662) or that given in the Explanatory Supplement. Here we describe a method devised in 1876, which first appeared in Butcher's Ecclesiastical Calendar, and which is valid for all years from 1583 onwards. It makes repeated use of the result of dividing one number by another number, the integer part being treated separately from the remainder. A calculator displays the result of a division as a string of numbers either side of a decimal point. The numbers appearing before (i.e. to the left of) the decimal point constitute the integer part; the decimal point and the numbers after (i.e. to the right of) the decimal point constitute the fractional part. The remainder may be found from the latter (including the leading decimal point) by multiplying it by the divisor (i.e. the number that you divided by) and rounding the result to the nearest integer value. For example, 2000/19 = 105.2631579. The integer part is 105, and the fractional part is 0.2631579. Multiplying the latter by 19 gives 5.000000100 so the remainder is 5.

We shall illustrate the method by calculating the date of Easter Day in the year 2009. This will give us practice for the sort of calculation we will be carrying out in the rest of this book.

Method				*Example*		
		Integer part	*Remainder*			
1.	Divide the year by 19.		a	$\frac{2009}{19}$	=	105.736 842 1
				a	=	14
2.	Divide the year by 100.	b	c	$\frac{2009}{100}$	=	20.090 000
				b	=	20
				c	=	9
3.	Divide b by 4.	d	e	d	=	5
				e	=	0
4.	Divide $(b+8)$ by 25.	f		f	=	1
5.	Divide $(b-f+1)$ by 3.	g		g	=	6
6.	Divide† $(19a+b-d-g+15)$ by 30.		h	$(19a+b-d-g+15)$	=	290
				h	=	20
7.	Divide c by 4.	i	k	i	=	2
				k	=	1
8.	Divide $(32+2e+2i-h-k)$ by 7.		l	l	=	1
9.	Divide $(a+11h+22l)$ by 451.	m		$(a+11h+22l)$	=	256
				m	=	0
10.	Divide $(h+l-7m+114)$ by 31.	n	p	$(h+l-7m+114)$	=	135
				n	=	4
				p	=	11
11.	The day of the month on which Easter Day falls is $p+1$.			$p+1$	=	12
	The month number is n (=3 for March, =4 for April).			n	=	4, so April
	Therefore Easter Day 2009 is					**12 April**

†$19a$ means 19 multiplied by a ($19 \times 14 = 266$ in this example).

The spreadsheet for this calculation, called DOE (the acronym for Date Of Easter), is shown in Figure 1.

It makes repeated use of two spreadsheet functions, TRUNC and MOD. (These are all examples of built-in, or intrinsic, spreadsheet functions; we will make use of many of the useful ones throughout this book.) The former truncates the number at the decimal point, so gives you the integer part of the number. Thus TRUNC(23.445) is 23. In cell C8 of the spreadsheet, the formula =TRUNC(C3/100) takes the number from cell C3 (2009 in this case), divides it by 100, and returns the integer part of the result (20).

The MOD function has two arguments separated by a comma. (An argument is a number or a reference within the brackets immediately following the function name. Two or more arguments are separated by commas.‡) The first argument is divided by the second argument, and then the remainder of the result is returned. Thus MOD(13,5) is 3 since 5 goes into 13 twice (2×5 = 10) leaving a remainder of 3 (i.e. 13−10). In cell C7 of the spreadsheet, the formula =MOD(C3,19) takes the number from cell C3 (2009 in this case), divides it by 19, and returns the remainder (14).

We have used the spreadsheet function IF in cell H4 to replace the month number, 3 or 4, with its name equivalent, ‘March’ or ‘April’. The IF function takes three arguments. The first is the test argument, which can be ‘true’ or ‘false’. In this case, the test argument is C22=3, i.e. if the number in cell C22 is equal to 3 the result of the test is ‘true’, and if not it is ‘false’. In this case, the number in cell C22 is 4 so the test returns ‘false’. The IF function returns the second argument (March in this case) if the test returns ‘true’, or the third argument (April in this case) if the test returns ‘false’, as here.

‡*Some spreadsheet programs use different separators; check yours.*

	A	B	C	D	E	F	G	H	I	J	K
1	**Date of Easter**										
2											
3	*Input*	**year**	**2009**			*Output*	**day**	**12**	=C21		
4							**month**	**April**	=IF(C22=3,"March","April")		
5							**year**	**2009**	=C3		
6											
7	1	*a*	14	=MOD(C3,19)							
8	2	*b*	20	=TRUNC(C3/100)							
9	3	*c*	9	=MOD(C3,100)							
10	4	*d*	5	=TRUNC(C8/4)							
11	5	*e*	0	=MOD(C8,4)							
12	6	*f*	1	=TRUNC((C8+8)/25)							
13	7	*g*	6	=TRUNC((C8-C12+1)/3)							
14	8	*h*	20	=MOD((19*C7)+C8-C10-C13+15,30)							
15	9	*i*	2	=TRUNC(C9/4)							
16	10	*k*	1	=MOD(C9,4)							
17	11	*l*	1	=MOD(32+2*(C11+C15)-C14-C16,7)							
18	12	*m*	0	=TRUNC((C7+(11*C14)+(22*C17))/451)							
19	13	*n*	4	=TRUNC((C14+C17-(7*C18)+114)/31)							
20	14	*p*	11	=MOD(C14+C17-(7*C18)+114,31)							
21	15	day	12	=C20+1							
22	16	month	4	=C19							

You can put any year after 1582 you like into cell C3 of the spreadsheet in place of 2009 and the date of Easter Day for that year will be calculated for you automatically. Try 2012. The answer should be 8 April.